Task a)

In[2348]:=

```
ClearAll["Global`*"]
f[x_, y_, α_] := y;
g[x_, y_, α_] := -Sin[x] - α * y;

(*Fixed points*)
sol = Solve[{f[x, y, α] == 0, g[x, y, α] == 0}, {x, y}]

(*Stability matrix*)
J[x_, y_, α_] = D[{f[x, y, α], g[x, y, α]}, {{x, y}}]
```

Out[2351]=

$\left\{\left\{y \to 0,\ x \to \boxed{2\pi\, c_1 \text{ if } c_1 \in \mathbb{Z}}\right\},\ \left\{y \to 0,\ x \to \boxed{\pi + 2\pi\, c_1 \text{ if } c_1 \in \mathbb{Z}}\right\}\right\}$

Out[2352]=

{{0, 1}, {-Cos[x], -α}}

From the task (and periodicity of the system) we know that we can limit our search to -π to π. Thus choose the constant c1 = -1,0 and 1. Moreover we study at the second solution to the system: $y \to 0,\ x \to \boxed{\pi + 2\pi\, c_1 \text{ if } c_1 \in \mathbb{Z}}$, since this is included in the interval.

x = -π

In[2353]:=

```
eig = Eigenvalues[J[-π, y, α]]

τ[x_, y_, α_] = eig[[1]] + eig[[2]] // FullSimplify
Δ[x_, y_, α_] = eig[[1]] * eig[[2]] // FullSimplify
(*Or*)
τAlt[x_, y_, α_] = Tr[J[-π, y, α]];
ΔAlt[x_, y_, α_] = Det[J[-π, y, α]];

(* Δ < 0 => saddle point regardless of alpha.*)
```

Out[2353]=

$\left\{\frac{1}{2}\left(-\alpha - \sqrt{4 + \alpha^2}\right),\ \frac{1}{2}\left(-\alpha + \sqrt{4 + \alpha^2}\right)\right\}$

Out[2354]=

$-\alpha$

Out[2355]=

-1

$x = 0$

In[2358]:=

```
Clear[eig, curve, eigVec]
eig = Eigenvalues[J[0, y, α]]
eigVec = Eigenvectors[J[0, y, α]]
τ[x_, y_, α_] = eig[[1]] + eig[[2]] // FullSimplify
Δ[x_, y_, α_] = eig[[1]] * eig[[2]] // FullSimplify
```

Out[2359]=

$$\left\{\frac{1}{2}\left(-\alpha-\sqrt{-4+\alpha^2}\right), \frac{1}{2}\left(-\alpha+\sqrt{-4+\alpha^2}\right)\right\}$$

Out[2360]=

$$\left\{\left\{\frac{1}{2}\left(-\alpha+\sqrt{-4+\alpha^2}\right), 1\right\}, \left\{\frac{1}{2}\left(-\alpha-\sqrt{-4+\alpha^2}\right), 1\right\}\right\}$$

Out[2361]=

$-\alpha$

Out[2362]=

1

we have $\Delta = 1 > 0$:

when $\alpha = 0$ => center

when $\alpha > 0$ => quadrant 4

Investigate the curve τ^2 - 4Δ to determine the different stability cases for different α:

In[2363]:=

```
curve[x_, y_, α_] = τ[x, y, α]^2 - 4 * Δ[x, y, α]
```

Out[2363]=

$-4 + \alpha^2$

If $0 < \alpha < 2$ we have a stable spiral. (Complex conjugate pair)

If $\alpha = 2$, we have a stable star or stable degenerate node (Since $\alpha \geq 0$). Moreover, for $\alpha = 2$ we have $\lambda 1 = \lambda 2$ and only one eigenvector => we have stable degenerate node.

If $\alpha > 2$ we have a stable node

$x = \pi$

In[2364]:=

```
Clear[eig, curve]
eig = Eigenvalues[J[π, y, α]];

τ[x_, y_, α_] = eig[[1]] + eig[[2]] // FullSimplify
Δ[x_, y_, α_] = eig[[1]] * eig[[2]] // FullSimplify
```

Out[2366]=

$-\alpha$

Out[2367]=

-1

Same τ and Δ as $x = -\pi$ = > saddle point

In summary:

In[2368]:=

```
Grid[{{"Fixed point", "α", "Type of fixed point"},
  {{-π, 0}, "any α since Δ < 0", "Saddle point"},
  {{0, 0}, "0 ", "Center"},
  {{0, 0}, "0 < α < 2 ", "Stable spiral"},
  {{0, 0}, "2 ", "Stable degenerate node"},
  {{0, 0}, "α > 2 ", "Stable node"},
  {{π, 0}, "any α since Δ < 0", "Saddle point"}},
 Spacings → {3, 3, 3},
 Dividers → All]
```

Out[2368]=

Fixed point	α	Type of fixed point
{-π, 0}	any α since Δ < 0	Saddle point
{0, 0}	0	Center
{0, 0}	0 < α < 2	Stable spiral
{0, 0}	2	Stable degenerate node
{0, 0}	α > 2	Stable node
{π, 0}	any α since Δ < 0	Saddle point

Task b)

We know that for the fixed points ($\pm\pi$,0), we can choose any alpha, they are saddle points regardless.

For the fixed point (0,0) We look at α =0, 1, 2, 4, such that we cover one alpha from each interval, α = 0, 0 < α < 2, α = 2 and α > 2.

```
ClearAll["Global`*"]
f[x_, y_, α_] := y;
g[x_, y_, α_] := -Sin[x] - α * y;

(*Parameters for the phase portrait plot.*)
minx = -π - π / 4;
maxx = π + π / 4;
```

```
miny = -3;
maxy = 3;

(*Fixed points from a)*)
fp = {{-π, 0}, {0, 0}, {π, 0}};

getInitialConditions[α_] := If[α == 0,
   Join[
    Table[{x, -3}, {x, minx, maxx, π / 6}],
    Table[{x, 0}, {x, -π + π / 6, π - π / 6, π / 6}],
    Table[{x, 3}, {x, minx, maxx, π / 6}],
    Table[{minx, y}, {y, miny, maxy, 0.2}],
    Table[{maxx, y}, {y, miny, maxy, 0.2}]],
   (*Initial conditions for alpha≠0*)
   Join[
    Table[{x, -3}, {x, minx, maxx, π / 6}],
    Table[{x, 3}, {x, minx, maxx, π / 6}],
    Table[{minx, y}, {y, miny, maxy, 0.5}],
    Table[{maxx, y}, {y, miny, maxy, 0.5}]]];

sol[x0_, y0_, α_] := NDSolve[{
    x'[t] == f[x[t], y[t], α],
    y'[t] == g[x[t], y[t], α],
    x[0] == x0,
    y[0] == y0},
   {x, y}, {t, 0, 12}];

(*Alphas determined from task a)*)
alphas = {0, 1, 2, 4};

plots = Table[Module[{α = a, p, s, initialC},
    (*Different initial conditions for α == 0*)
    initialC = getInitialConditions[α];
    p = Show[Table[ParametricPlot[Evaluate[
          {x[t], y[t]} /. sol[initialC[[i, 1]], initialC[[i, 2]], α]], {t, 0, 12},
         PlotRange → {{minx, maxx}, {miny, maxy}},
         AxesLabel → {"x", "y"}] /.
        Line[x_] :→ {Arrowheads[{{0.025, 0.25}, {0.025, 0.5}}], Arrow[x]},
       {i, 1, Length[initialC]}], ListPlot[fp, PlotStyle → {Red},
       PlotMarkers → {Automatic, 8}], ImageSize → 600];

    (*Add title to parametric plot.*)
    Grid[{{Style["α = " <> ToString[a] <> ",  Fixed Points: " <> ToString[fp],
        FontSize → 14, Bold]},
      {p}}]],

   {a, alphas}
  ];

GraphicsColumn[plots, Spacings → 3, ImageSize → 800]
```

Out[2407]=

α = 0, Fixed Points: {{-Pi, 0}, {0, 0}, {Pi, 0}}



α = 1, Fixed Points: {{-Pi, 0}, {0, 0}, {Pi, 0}}





α = 2, Fixed Points: {{−Pi, 0}, {0, 0}, {Pi, 0}}



α = 4, Fixed Points: {{−Pi, 0}, {0, 0}, {Pi, 0}}




−3